
Урок № 3 

Работа со справочниками 

 
Очень часто при оформлении таблиц и при проведении расчетов 

используются данные, которые целесообразно оформлять в виде 

отдельных таблиц-справочников. К таким данным относятся, 

например, цены в каталоге, ставки налога в налоговой таблице, 
тарифы на телефонные переговоры, курсы валют и др. Некоторые  

данные можно рассматривать в виде списка значений, когда 

определяющим для выбора нужного значения является его номер в 
списке. Например, названия месяцев года, стоимость доставки груза в 

зависимости от номера тарифной зоны, квартальные объемы выпуска 

продукции  и др. 

При построении формул, когда требуется обращение к таблице 
со справочными данными или выбор значения из списка, 

используются функции ВЫБОР, ВПР,  ГПР и  ПОИСКПОЗ из категории 

функций «Ссылки и массивы». 
 

 

1. Функция  ВЫБОР 

 

Функция ВЫБОР имеет следующий синтаксис: 

ВЫБОР(номер_индекса;значение1;значение2;…) 

Она возвращает значение элемента из списка, задаваемого 
аргументами, начиная со второго («значение1», «значение2» и т.д.), по 

его номеру («номер_индекса»)  в списке. 

 При работе с функцией ВЫБОР необходимо учитывать следующее: 

1. Аргументы могут быть заданы константами, ссылками на клетки, 
выражениями, но первый аргумент должен принимать значения из 

множества натуральных чисел.  

2. Если первый аргумент принимает значение не их множества 

натуральных чисел или  натуральное значение, большее количества 
значений в списке, то функция возвращает ошибочное значение 

#ЗНАЧ!. 

Пример 1. В зависимости от содержимого клетки A1 формула  

=ВЫБОР(A1;”янв-март”;”апр-июнь”;”июль-сент”;”окт-дек”) 
 

возвращает  следующие результаты: 
 

A1 
Возвращаемое 

значение 
Пояснение 

3 июль-сент  

0 #ЗНАЧ! 
Недопустимый номер элемента 

12 #ЗНАЧ! 

 

 

Пример 2. Имеется информация о 

стоимости входного билета в 

музей, размещенная в ЭТ так, как 
показано на рис. 1. Если в клетку 

D1 ввести индекс категории 

посетителя музея, то узнать 
стоимость билета можно с 

помощью формулы: 
 

=ВЫБОР(D1;C2;C3;C4;C5;C6) 
 

 

Пример 3. Существуют различные варианты начисления процентов 
на банковские вклады (депозиты). Чаще всего встречаются 

следующие три варианта:  

1. Начисление процентов в конце срока по депозиту; 
2. Ежеквартальное начисление процентов (и капитализация 

процентов ежеквартально); 

3. Ежемесячное начисление процентов (и капитализация 
процентов ежемесячно). 

Исходными данными для определения размера вклада являются: 
вариант начисления процентов (задается номером варианта из 

приведенного списка, содержимое клетки A1), сумма депозита 

(содержимое клетки A2), годовая процентная ставка на депозит 

(содержимое клетки A3,  значение вводится в формате «процент», т.е.,  
например, 10%). 

Размер банковского вклада на конец первого года можно 

определить по формулам: 
 

=ВЫБОР(A1;A2*(1+A3);A2*(1+A3/4)^4;A2*(1+A3/12)^12) 

Рис. 1. 

или 

=A2*ВЫБОР(A1;1+A3;(1+A3/4)^4;(1+A3/12)^12) 

 
Пример 4. В ЭТ построена таблица правильных ответов на задачи  и 

количества баллов за их правильное решение (рис. 2).   

 

 
 

Рис. 2 

 

Пусть в клетку A10 вводится номер задачи, а в клетку A11 – ответ на 
задачу, полученный студентом, тогда для определения количества 

заработанных баллов можно воспользоваться формулой: 
 

=ЕСЛИ(A11=ВЫБОР(A10;B2;C2;D2;E2;F2); 

ВЫБОР(A10;B3;C3;D3;E3;F3);0) 
 

 Очевидно, что, если количество значений в списке велико, или 

аргументы, формирующие список, задаются длинными строками, то 

не разумно использовать функцию  ВЫБОР.  Для извлечения данных 

из списка и справочных таблиц  гораздо удобнее воспользоваться 

функциями ВПР и  ГПР. 

 

2. Функции  ВПР и  ГПР 

 

Функции ВПР и  ГПР имеют следующий синтаксис: 

ВПР(искомое_значение; таблица; номер_столбца; [интервальный_ 
просмотр]), 

ГПР(искомое_значение; таблица; номер_строки; [интервальный_ 
просмотр]). 

Аргументы функций имеют следующий смысл. 

Искомое_значение – это значение, которое требуется найти в первом 
столбце  «таблицы» (для ВПР) или в первой строке «таблицы» (для ГПР). 

«Искомое_ значение» может быть константой, ссылкой  на клетку или 

выражением. 

Таблица – это диапазон на листе ЭТ, в котором ищутся данные. 
Диапазон может быть задан адресом или именем.  

Номер_столбца (Номер_строки) – номер столбца (строки) в  диапазоне 

«таблица», из которого (которой) возвращается требуемое значение.  

Нумерация строк и столбцов  в «таблице» начинается с 1. 
Интервальный_просмотр

  – необязательный аргумент, который 

определяет, нужно ли, чтобы функция ВПР (ГПР) искала точное (если 

он принимает значение ИСТИНА или 1) или приближенное 
соответствие (если аргумент принимает значение ЛОЖЬ или 0).  

 При работе с функциями ВПР и  ГПР необходимо учитывать 

следующее: 

1. Первый столбец (первая строка) диапазона «таблица» должны 

содержать значения одного  и  того  же  типа. 
2. Если аргумент «Интервальный_просмотр» имеет значение ИСТИНА 

(или 1), то данные первого столбца (строки) должны быть 

отсортированы по возрастанию.  
3. Если аргумент «Интервальный_просмотр» имеет значение ИСТИНА 

(или 1), то функция ВПР (ГПР) ищет среди значений первого 

столбца (строки) ближайшее значение, меньшее или равное 

«искомому значению» (режим диапазонного поиска).  
4. Если аргумент «Интервальный_просмотр» имеет значение ЛОЖЬ 

(или 0), то функция ВПР (ГПР) ищет среди значений первого 

столбца (строки) значение, равное «искомому значению» (режим 
точного поиска).  

5. Если аргумент «Интервальный_просмотр» опущен, то, по умолчанию, 

действует режим диапазонного поиска. 
6. Если «искомое_значение» меньше, чем наименьшее значение в 

первом столбце (строке) «таблицы», то функция ВПР (ГПР) 

возвращает значение ошибки #Н/Д. 

7. Если  «Интервальный_просмотр» имеет значение ЛОЖЬ, а среди 
значений первого столбца (строки) диапазона «таблица» нет 

«искомого значения», то функция ВПР (ГПР) возвращает ошибочное 

значение #Н/Д. 
8. Если аргумент «номер столбца» («номер строки») принимает 

значение, большее, чем количество столбцов (строк) в диапазоне 

«таблица», то функция ВПР (ГПР) возвращает значение ошибки 
#ССЫЛКА!.  

9. Если аргумент «номер столбца» («номер строки») принимает 

значение, меньшее 1, то функция ВПР (ГПР) возвращает значение 

ошибки #ЗНАЧ!.  

Таким образом, основное различие между функциями  ВПР и  ГПР 

заключается в направлении просмотра при поиске заданного значения 

(рис. 3). 

Рис. 3. Для функции ВПР поиск определяет строку (а),  

а для ГПР – столбец (б) 

 

 

3. Пример организации справочника 
 

Имеется информация о налоговых ставках для легковых автомобилей, 
установленных в 2012 и 2013 годах (табл. 1).   

 
Табл. 1. Налоговые ставки для легковых автомобилей 

Мощность двигателя 

Налоговая 

ставка (руб) 

в 2012 году 

Налоговая 

ставка (руб) 

в 2013 году 

до 70 л.с. включительно 0 0 

свыше 70 л.с. до 100 л.с. 
включительно  

7 12 

свыше 100 л.с. до 125 л.с. 

включительно  
20 25 

свыше 125 л.с. до 150 л.с. 

включительно  
30 35 

свыше 150 л.с. до 175 л.с. 

включительно  
38 45 

свыше 175 л.с. до 200 л.с. 

включительно  
45 50 

свыше 200 л.с. до 225 л.с. 

включительно  
60 65 

свыше 225 л.с. до 250 л.с. 
включительно  

75 75 

свыше 250 л.с.  150 150 

 

 
Для расчета транспортного 

налога необходимо в ЭТ 

подготовить справочник, 

заполнив  первую колонку 
так, как показано на рис. 4. 

Интервалы значений 

мощности следует задавать их 
нижними границами. Считая, 

что мощность определяется с 

точностью  1 л.с., для первого 

интервала (до 70 л. с. 
включительно) нижняя 

граница равна 0, для второго 

(свыше 70 л. с. до 100 л. с. 
включительно) – 71, и т. д. 

Если в клетку A15 ввести мощность двигателя  (в л. с.), а в клетку A16 

– год (2012 или 2013), то с помощью выражения: 
 

=ВПР(A15;A3:C11;ЕСЛИ(A16=2012;2;3)) 
 

можно определить размер налоговой ставки.  
 

 

  

Рис. 4. Справочник  налоговых ставок 

4. Функция  ПОИСКПОЗ 

 
Очень часто справочник содержит более двух колонок (строк), и при 

этом номер колонки (строки), из которой надо извлечь информацию, 

определяется дополнительными условиями. В рассмотренном выше 

примере номер колонки зависит от года и для его определения 
использовалась функция ЕСЛИ.  

Для работы со справочниками в Excel бывает полезной 

функция, ПОИСКПОЗ которая позволяет определять номер заданного 
элемента в списке. Функция ПОИСКПОЗ имеет следующий синтаксис: 
 

ПОИСКПОЗ(искомое_значение,просматриваемый_массив, 
[тип_сопоставления]) 

 

Искомое_значение – заданное значение, которое сопоставляется со 

значениями в списке «просматриваемый_массив». «Искомое_ значение» 

может быть константой, ссылкой  на клетку или выражением. 
 

«Просматриваемый_массив» – это диапазон ячеек, в которых 
производится поиск. Диапазон может быть задан адресом или именем.  

 

«Тип_сопоставления» – это необязательный аргумент. Он указывает на 

то, каким образом искомое значение сопоставляется со значениями 

просматриваемого массива. Аргумент может принимать  значения -1, 
0 или 1. Если аргумент принимает значение -1, то при просмотре 

ищется ближайшее значение,  большее или равное  искомому. Если 0, 

то ищется значение, в точности совпадающее с искомым. Если  1, то  
ищется ближайшее значение, меньшее или равное искомому. Если 

аргумент опущен, то, по умолчанию, для него  используется значение 

1. 
 

 При работе с функцией  ПОИСКПОЗ следует учитывать следующее: 

1. ПОИСКПОЗ не различает регистры при сопоставлении текстовых 
значений (значения «валюта» и «ВАЛЮТА» эквиваленты). 

2. Если аргумент «Тип_сопоставления» имеет значение 1 или 

опущен, то значения в диапазоне «просматриваемый_массив» 
должны быть упорядочены по возрастанию. 

3. Если аргумент «Тип_сопоставления» имеет значение -1, то 

значения в диапазоне «просматриваемый_массив» должны быть 

упорядочены по убыванию. 

4. Если аргумент «Тип_сопоставления» имеет значение 0, а  искомого 

значения нет в просматриваемом массиве, то функция ПОИСКПОЗ 

возвращает ошибочное  значение  #Н/Д. 
5. Если аргумент «Тип_сопоставления» имеет значение 1 (или -1), а  

искомое значение меньше  (больше) первого элемента в 

просматриваемом массиве, то функция ПОИСКПОЗ возвращает 
ошибочное  значение  #Н/Д. 

 

 

5. Примеры использования  функций ВПР, ГПР и ПОИСКПОЗ 

 

Имеются данные о курсе 
различных валют к 

российскому рублю
1
, 

подготовленные в виде 

таблицы на листе Курс 

(рис. 5).  Приведем 

примеры использования 

функций ВПР и 
ПОИСКПОЗ для 

извлечения курса 

валюты по заданной 

дате. 
 

Пример 1. Пусть в 

клетку  A1 (не на листе 
Курс) вводится дата, 

тогда с помощью 

выражения  

=ВПР(A1;Курс!A3:E16;2) 

можно установить курс доллара США на введенную дату. Так как не 

указан четвертый аргумент, то действует правило диапазонного 

поиска (в первой колонке указанного диапазона ищется дата, 

совпадающая с заданной в клетке A1, а, если такой нет, – ближайшая 
к ней, но более ранняя). В зависимости от содержимого клетки A1 

будут получены следующие результаты: 

                                                
1
 Источник данных:   http://bankir.ru/kurs/  

Рис. 5. Лист Курс 

http://bankir.ru/kurs/

A1 
Курс 

доллара США 
Пояснение 

20.07.2013 32,4288  

28.07.2013 32,6371 

Даты в списке дат нет, поэтому курс 

определяется по ближайшей более ранней 

дате, т.е. 27.07.2013. 

15.07.2013 #Н/Д 
В списке дат нет указанной даты и более 

ранней к ней. 

12.08.2013 32,9390 

Указанной даты в списке дат нет, но 

ближайшая более ранняя к ней, это – 

7.08.2013 

 

Пример 2. Пусть в клетку  A1 (не на листе Курс) вводится дата, тогда с 
помощью выражения  

=ВПР(A1;Курс!A3:E16;3;0) 

можно установить курс евро на дату, точно совпадающей с датой, 

введенной в клетку A1. В зависимости от содержимого клетки A1 

будут получены следующие результаты: 
 

A1 Курс евро Пояснение 

20.07.2013 42,5920  

28.07.2013 #Н/Д 

В списке дат нет указанной даты 15.07.2013 #Н/Д 

12.08.2013 #Н/Д 

 

Проверить, возвращает ли функция ошибочное значение, можно с 
помощью  функции ЕОШИБКА: 

=ЕОШИБКА(ВПР(A1;Курс!A3:E16;3;0)) 

Например, в зависимости от содержимого клетки A1 получим: 

A1 
Результат 

проверки 

20.07.2013 ЛОЖЬ 

28.07.2013 ИСТИНА 

 
Пример 3. Пусть в клетку  A1 вводится дата, а в клетку A2 – 

произвольное натуральное число (обе клетки не на листе Курс), тогда с 

помощью выражения  

=ВПР(A1;Курс!A3:E16;A2;ИСТИНА) 

можно установить курс валюты, которая соответствует колонке с 

номером, определяемым содержимым клетки A2,  на дату, точно 

совпадающей с датой, введенной в клетку A1, или ближайшей к ней. В 
зависимости от содержимого клеток A1 и A2 будут получены 

следующие результаты: 
 

A1 A2 
Курс 

валюты 
Пояснение 

20.07.2013 4 49,4020 Курс фунтов стерлингов 

28.07.2013 3 43,3421 

Даты в списке дат нет, поэтому 

курс евро определяется по 

ближайшей более ранней дате, т.е. 

27.07.2013. 

15.07.2013 4 #Н/Д 
В списке дат нет указанной даты и 

более ранней к ней. 

12.08.2013 7 #ССЫЛКА 
Недопустимый номер колонки, так 
как диапазон Курс!A3:E16 

содержит всего 5 колонок  

19.07.2013 -2 #ЗНАЧ! 
Недопустимый номер колонки. 

Значение аргумента меньше 1. 

 

Пример 4. Пусть в клетку A1 вводится название валюты, тогда с 

помощью формулы 

=ПОИСКПОЗ(A1;Курс!B2:E2;0)+1 

можно определить, в какой колонке диапазона Курс!A3:E16 

содержится информация о курсе валюты, название которой точно 

совпадает с содержимым клетки A1.  В зависимости от содержимого 

клетки A1 будут получены следующие результаты: 
 

A1 
Номер 

колонки 
Пояснение 

Доллар США 2  

Доллар #Н/Д В списке названий валют нет указанного 

евро 3  

 

Пример 5. Пусть в клетку A1 вводится дата, тогда с помощью 

формулы 

=ПОИСКПОЗ(A1;Курс!A3:A16;1) 

можно определить, какая строка диапазона Курс!A3:E16 содержит 

данные о курсе валют на введенную дату.  В зависимости от 

содержимого клетки A1 будут получены следующие результаты: 
 

A1 
Номер 

строки 
Пояснение 

17.07.2013 #Н/Д 
В диапазоне нет ни совпадающей,   ни 
более ранней даты по отношению к 

заданной 

25.07.2013 5  

28.07.2013 7 
Номер элемента определяется по 

ближайшей дате 27.07.2013 

10.08.2013 14 
Номер элемента определяется по 

ближайшей дате 7.08.2013 

 

 

Если при проведении расчетов нежелательно получать результат в 
виде ошибочного значения «#Н/Д», то можно построить выражение с 

использованием логической функции  ЕСЛИ:   

=ЕСЛИ(ЕОШИБКА(ПОИСКПОЗ(A1;Курс!A3:A16;1));”Нет данных”; 
ПОИСКПОЗ(A1;Курс!A3:A16;1)) 

Теперь вместо ошибочного значения будет возвращаться текст «Нет 

данных». 

Пример 6. Пусть в клетку  A1 вводится дата, а в клетку A2 – название 

валюты (обе клетки не на листе Курс), тогда с помощью выражения  

=ВПР(A1;Курс!A3:E16;ПОИСКПОЗ(A2; Курс!B2:E2;0)+1;ИСТИНА) 

можно установить курс заданной валюты на введенную дату. В 

зависимости от содержимого клеток A1 и A2 будут получены 
следующие результаты: 
 

A1 A2 
Курс 

валюты 
Пояснение 

20.07.2013 евро 42,5920  

15.07.2013 
латвийский 

лат 
#Н/Д 

В списке дат нет указанной даты и 

более ранней к ней. 

17.07.2013 
Датская 

крона 
#Н/Д Заданной валюты нет в списке 

 

Можно составить формулу для определения курса валюты на 
заданную дату и с использованием функции ГПР: 



12 

 

=ГПР(A2;Курс!B2:E16;ПОИСКПОЗ(A1; Курс!A3:A16)+1;ЛОЖЬ) 

 

Пример 8.  Формулу в примере 4 из пункта «Функция ВЫБОР» 
можно заменить  на следующую: 

=ЕСЛИ(A11=ГПР(A10;B1:F3;2;0); ГПР(A10;B1:F3;3;0);0) 

Пример 9.  Имеются справочные данные о расстоянии (в км) от г. 

Петрозаводск до городов России, в которые осуществляется доставка 

посылок весом до 500 г наземным транспортом, и тарифы на 

пересылку в зависимости от расстояния до места назначения. Данные 
оформлены в виде двух справочников (рис. 6). 
 

 
Рис. 6 

 

Узнать тариф на пересылку посылки в заданный город (пусть, 

например название города введено в клетку A100) можно с помощью 
формулы: 

=ВПР(ВПР(A100;A4:B16;2;0);E5:F9;2) 

Обратите внимание на то, что область  справочника «Тарифы»  

определяет диапазон  E5:F9 (см. п 3).  

 

 

6. Выбор значения из списка 
 
Чтобы избежать ошибок при вводе данных в клетку, можно 

установить для клетки режим ввода с выбором значения из списка. 

Рассмотрим, как это сделать на следующем примере.  

 Введите в клетку A1 текст «Валюта». А в клетку A2 будем вводить 

название валюты из списка: доллар США, евро, фунт стерлингов, 

латвийский лат. Чтобы назначить режим выбора названия валюты из 

списка, выполните следующие действия: 

1. Сделайте клетку A2 активной
2
. 

2. Выберите на вкладке Данные команду Проверка данных. 

3. В открывшемся диалоговом окне «Проверка вводимых 

значений» (рис. 7) измените значение поля «Тип данных», 
выбрав из раскрывающегося списка значение «Список» (рис. 

8) 

4. Задайте значение поля «Источник», введя названия валют, 

разделяя их точкой  с запятой (рис. 9) 
5. Закройте диалоговое окно, нажав кнопку OK. 

 

 
 
 

Рис. 7. Диалоговое окно «Проверка вводимых значений» 

 

 

                                                
2
 См. Замечание 3, приведенное в конце пункта. 

Рис. 8. Выбор значения поля «Тип данных» 

 
Теперь справа от выделенной клетки появилась кнопка 

«раскрывающегося списка» (рис. 10),  нажав на которую получим 

список названий валют, из которого и выбираем требуемое название, 
например,  «доллар США» (рис. 11). 

 

 
 

Рис. 9. Заполнение поля «Источник» - задание элеменов списка 

 

Если список содержит большое количество значений, то лучше 
заранее подготовить в таблице диапазон с данными (строку или 

столбец), введя каждое значение в отдельную клетку. И тогда, 

заполняя поле «Источник» следует ввести адрес диапазона, 

содержащий список значений
3
. Например, если бы был подготовлен 

лист Курс (рис. 5), то адрес диапазона с данными  был бы такой, как 
показано на рис. 12. 

 

 
 

Рис. 10. Кнопка «раскрывающегося списка» 

 

 

 

  
 

Рис. 11. Выбор названия валюты из списка 

 
 

 

 
 

Рис. 12. Заполнение поля «Источник» - задание диапазона с данными 

                                                
3
 См. замечание 1, приведенное в конце пункта «Выбор значения из списка». 

Замечание 1.  Не все версии Microsoft Excel поддерживают 

возможность при заполнении поля «Источник»   ссылаться на 

диапазон другого листа с данными об элементах списка.  В этом 
случае или диапазон со списком следует размещать на листе, где 

находятся клетки,  для которых устанавливается режим с проверкой. 

Или, что предпочтительнее,  диапазону с элементами списка следует 
дать имя (см. пункт «Именование диапазонов») и использовать его 

при заполнении поля «Источник».   

Замечание 2. Установленный для клетки режим не отменяет 
возможности ввести значение обычным образом, только при этом 

введенное в ручном режиме значение «подвергается» проверке на 

принадлежность списку. Если введенное значение не содержится в 

списке, то выдается стандартное сообщение об ошибке. Работая с 
вкладками «Сообщение для ввода» и «Сообщение об ошибке» 

диалогового окна «Проверка вводимых значений», можно изменить 

текст сообщения об ошибке и, кроме того, добавить свой текст 
сообщения для ввода. 

Замечание 3. Режим ввода с выбором значения из списка 

устаналивается для выделенного диапазона клеток.  

 

 

7. Именование диапазонов 
 

 Дать имя выделенному диапазону можно одним из 

следующих способов: 
1. Переместив курсор в поле имени (Урок № 1, рис. 1), ввести в 

нем желаемое имя диапазона и закончить ввод нажатием 

клавиши Enter (  ).   

2. Вызвав правой кнопкой мыши контекстное меню, выбрать 

пункт «Присвоить имя» и заполнить поле «Имя» диалогового 

окна «Создание имени». 
3. Выполнить команду Присвоить имя на вкладке Формулы. 
 
Список всех заданных имен диапазонов можно вывести, если 
щелкнуть левой кнопкой мыши на значке  в правой части поля 

имени. 

8. Задача «Расчет платы за проживание в гостинице» 
 
Постановка задачи 

Требуется подготовить ЭТ, с помощью которой можно выполнить 

расчет платы за проживания в гостинице, которая зависит от типа 
номера  (однокомнатный, люкс, двухкомнатный) и срока проживания. 

Известна стоимость за один день проживания в номере, измеряемая в 

долларах. Сумма к оплате должна рассчитываться в рублях, зная курс 
доллара. Установлена скидка (в процентах) на стоимость одного дня 

проживания в случае, если срок проживания в номере составляет не 

менее заданного количества дней.  

 Расчетная таблица платы за проживание в гостинице и 

необходимые для расчета справочные данные должны размещаться на 

разных листах книги. 
 

Подготовка справочников 

Дайте первому листу книги название Справочники. И подготовьте на 
нем справочные таблицы так, как показано на рис. 13. 

 

 
 

Рис. 13. Таблицы со справочными данными 

Выполните именования клеток и диапазонов в соответствии с 

таблицей: 
 

Клетка или диапазон Имя 

Справочник!A2:A16 Номера_комнат 

Справочник!D2:D4 Типы_комнат 

Справочник!D2:E4 Цены 

Справочник!H1 Курс_доллара 

Справочник!H3 Скидка 

Справочник!H4 Дней_не_менее 

 
Подготовка журнала 

Дайте второму листу книги название Расчет. Подготовьте на этом 

листе таблицу в таком виде, как показано на рис. 14.  Установите 
необходимый режим выравнивания заголовков и выделите цветом 

столбцы, которые будут содержать формулы (C, D и F-G). 

 

 
 

Рис. 14. Заголовок таблицы  

 

Установите для диапазона C3:C9 режим ввода номера комнаты из 

«раскрывающего списка». Источник значений для списка   диапазон 
Номера_комнат. 
 

Заполнение журнала 
 

1. Задание исходных данных 

Заполните колонки A, B и C  исходными данными (рис. 15). 

Рис. 15. Исходные данные к расчету 

 

Установите для клеток диапазонов A3:A9, C3:C9 и E3:E9 формат 
горизонтального выравнивания «по центру».  

 

2. Заполнение колонки D 
 

Зная номер комнаты для первого клиента, ее тип определим по 
справочнику с помощью функции ВПР: 

=ВПР(C3; Справочник!A2:B16;2;0) 

где диапазон  A2:B16 на листе Справочники задает область 

справочных данных.  

Для всех остальных клиентов при определении типа комнаты в 

формуле следовало бы изменить ссылку на соответствующую клетку с 

номером комнаты клиента, диапазон же со справочными данными 
оставить без изменения. Поэтому, чтобы правильно заполнить 

колонку D копированием формулы из клетки D3 на диапазон D4:D9, 

надо отредактировать формулу следующим образом: 

=ВПР(C3; Справочник!A$2:B$16;2;0) 

(диапазон с абсолютизированными строками при копировании 

формулы не изменяется). 

 Введите формулу в клетку D3 и выполните ее копирование на 
диапазон  D4:D9.  Результат копирования показан на рис. 16. 

Рис. 16. Заполненная колонка D 

 
В формуле вместо адреса диапазона удобнее использовать его имя, 

например, Комнаты. Введите в клетку D3 формулу 

=ВПР(C3;Комнаты;2;0) 

В результате в клетке D3 будет отображено ошибочное значение  
#ИМЯ?  («Неизвестное имя»). Действительно, такого имени нет в 

списке имен диапазонов, созданных на этапе подготовки 

справочников. Задайте диапазону Справочник!A2:B16 имя Комнаты. 

После этого список имен в книге будет дополнен еще одним 

(посмотрите список!) и после этого будет получен желаемый 

результат в клетке D3 . Осталось только еще раз выполнить 
копирование  формулы из клетки D3  на диапазон  D4:D9. 

 

3. Заполнение колонок F, G и H 

Зная тип комнаты, стоимость проживания за 1 день (в долларах) 
определим из справочника цен  (диапазон Цены) с помощью функции 

ВПР. В клетку F3 введите формулу 

=ВПР(D3;Цены;2;0) 

Стоимость в рублях за 1 день проживания в комнате найдем, зная курс 

доллара (содержимое клетки Справочники!H1 с именем Курс_доллара). 

Для этого введите в клетку G3 формулу: 

=F3*Курс_доллара 

Для определения суммы к оплате  (в рублях)  в клетку  H3 введите 
формулу: 

=G3*E3 

Установив для клеток F3 и G3 формат горизонтального выравнивания  
«по центру» формат выравнивания «по центру»,  скопируйте 

содержимое диапазона  F3:H3 на диапазон F4:H9. 

Результат заполнения колонок таблицы показан на рис. 17. 
 

 

Рис. 17. Результат заполнения таблицы 

 

  Результаты построения  сохраните в книге с названием 

Гостиница. 
 

 Упражнение.  Предположим, что для  клиентов, проживающих 

в гостинице не менее заданного количества дней (например, 9) 

установлена скидка на стоимость за один день проживания 
(например, 5%). Как следует изменить формулы в колонке H?  

 

Контрольные вопросы 

1. Для чего используются функции ВЫБОР, ВПР, ГПР и  ПОИСКПОЗ? 

2. Должны ли быть упорядочены значения в первом столбце (первой 
строке) диапазона с данными при использовании функции ВПР (ГПР)? 

3. Синтаксис функций ВЫБОР, ВПР, ГПР и  ПОИСКПОЗ . 

4. В каких случаях функции  ВПР, ГПР и  ПОИСКПОЗ возвращают 

ошибочные значения  #Н/Д,  #ССЫЛКА!? 
5. Когда функция ВЫБОР  возвращает ошибочное значение #ЗНАЧ!? 

 


